
Introducción. 
 
Una simulación es el resultado de una serie de operaciones matemáticas que nos permitirá 
observar y estudiar el comportamiento de nuestro inductor para diferentes condiciones de 
trabajo, es de decir, diferentes tensiones aplicadas en sus terminales, ya sean estas 
constantes, como las de una batería, o variables en el tiempo, como las alternas senoidales. 
 
Para proceder con la simulación necesitaremos la descripción matemática del comportamiento 
de nuestro inductor. Detalle que podemos encontrar perfectamente detallado en libros sobre 
teoría de circuitos. 

Modelo Matemático en el Dominio del Tiempo. 
 
En este capítulo se describe el desarrollo matemático que permite analizar y simular el 
comportamiento de un circuito RL serie, de la forma más simple. En modelos más complicados 
se pueden incluir variables como la permeabilidad magnética, y las dimensiones del material 
ferromagnético, así como observar los efectos producidos por la saturación y los efectos 
producidos por la respuesta dinámica del material ferromagnético, obteniendo un modelo 
mucho más complicado, pero de mayor utilidad, tanto en el análisis como en el diseño de 
inductores. 
 
En el siguiente circuito, tenemos, una fuente de tensión SV , una resistencia SR  y un inductor 

L , conectados en serie. Donde, SV  es la tensión aplicada a nuestro inductor, RSV  es la 

tensión debida a la resistencia del hilo de cobre utilizado en el inductor SR , y LV  es la tensión 

en debida a la inductancia L . Dicho de otro modo, hemos modelado nuestro inductor real 
mediante una inductancia conectada en serie con un inductancia. 

Entonces, tenemos: 
 
Según la Ley de Corrientes de Kirchhoff ( LCK ): La suma algebraica de corrientes que entra en 
cualquier nodo es cero. 
 
En este caso: 
 
 0=+ RSVS ii  

 0=+ LRS ii  

 0=+ VSL ii  

Donde VSi  es la corriente de SV , RSi  es la corriente de SR  y Li  es la corriente de L . 
 
Según la Ley de Voltajes de Kirchhoff ( LVK ): La suma algebraica de voltajes alrededor de 
cualquier trayectoria cerrada en un circuito es cero. 
 
En este caso: 
 
 LRSS VVV +=         ( 1 ) 
 
Como: 
 

dt
tdiLtVL
)(

)( =  

)()( tiRtV SRS ⋅=  
 
De donde obtenemos: 
 

 )(
)( tiR

dt
tdiLV SS ⋅+=        ( 2 ) 

 
dt

tdiLtiRV SS
)(

)( =⋅−  

 
L

tiRV
dt

tdi SS )()( ⋅−=        ( 3 ) 

 
Entonces, como: 
 
 )()()( dttititdi −−=  para dt suficientemente pequeño. 
 
Tenemos que: 
 

 
L

tiRV
dt

dttiti SS )()()( ⋅−=−−
  

Por tanto: 

 
dtRL

dttiLdttVti
S

S

⋅+
−⋅+⋅= )()(

)(       ( 4 ) 

 
Esta expresión matemática nos permitirá determinar la corriente que circula por el circuito en un 
determinado instante de tiempo )(ti , a partir de la corriente en el instante anterior )( dtti −  y 

de la señal de entrada aplicada )(tVS . De forma que podemos implementar fácilmente un 
algoritmo iterativo con el que podremos simular el comportamiento del circuito indicado. Modelo 
que proporcionará un resultado suficientemente bueno para un dt  suficientemente pequeño. 

Por ejemplo, en SciLab podríamos utilizar el siguiente algoritmo ( rl.sci ):  
 

// Simulación Inductor ( Circuito RL Serie ) ( rl.sci ). 
 
clear 
 
Vs = 10;      // en Voltios cte en este caso 
 
L = 0.016;    // en Henrios 
Rs = 0.1;     // en Ohms 
 
Tau = L/Rs;     // Parámetros simulación 
tfin = 6*Tau;   // en s 
dt = Tau/10;    // en segundos 
 
i(1) = 0;     // Condiciones iniciales 
t(1) = 0;     // Condiciones iniciales 
VL(1)= Vs;    // per t=0, i=0, VR=0 => VL=Vs 
Vr(1)= 0; 
Pi(1)=0; 
 
for n = 2:tfin/dt 
  t(n) = t(n-1)+dt; 
  i(n) = (Vs*dt+L*i(n-1))/(L+Rs*dt); 
  Vr(n)= i(n)*Rs; 
  VL(n)= L * (i(n)-i(n-1))/dt; 
  Pi(n)= i(n)*Vs; 
end; 
 
scf(0) 
clf() 
subplot(3,1,1) 
plot(t,VL) 
xtitle('','t [s]','VL [V]') 
 
subplot(3,1,2) 
plot(t,i) 
xtitle('','t [s]','Current [A]') 
 
subplot(3,1,3) 
plot(t,Pi) 
xtitle('','t [s]','Pi [W]') 

Con lo que obtendremos el siguiente resultado:  

Para ajustar, automáticamente el tiempo de simulación en función de los parámetros, se 
determinan las variables fint  i dt  en función de los valores de SR  y de L , de la siguiente 

manera: 
 
Siendo: 
 

SR
L=τ  

 
Determinaremos: 
 

 τ6=fint  y  
10

τ=dt  

 
Con lo que nos aseguraremos una representación gráfica de los resultados correcta para 
cualquier valor de SR , y  de L . 

Efectos de Rs en el Dominio del Tiempo. 
 
Podemos observar el efecto de la variación de SR , L  y SV sobre el comportamiento del 
circuito realizando una serie de simulaciones para distintos valores de dichos parámetros y 
mostrarlos en una gráfica que no permita observar la diferencia. 
 
 
Por ejemplo, podemos modificar le algoritmo anterior para obtener el comportamiento del 
inductor cuando le aplicamos un señal de entrada senoidal descrita por la ecuación: 
 
 )()( tsenVtVs P ⋅⋅= ω  
 
También se ha realizado otra modificación que permite obtener el comportamiento del inductor 
a distintos valores SR , es conveniente recordar que este parámetro representa el diámetro y la 
longitud del hilo de cobre esmaltado. Ver anexo algoritmos, rl_Rs.sci. 
 
 
Obteniendo el siguiente resultado: 

Efectos de L en el Dominio del Tiempo. 
 
Para obtener el comportamiento del inductor para distintos valores L , se ha utilizado este otro 
algoritmo, ver anexo algoritmos rl_L.sci. Con el que se ha obtenido el siguiente resultado: 

Modelo Matemático en el Dominio de la Frecuencia. 
 
Para poder apreciar mejor el comportamiento del inductor a distintas frecuencias, lo mejor es  
buscar la descripción matemática en función de s , siendo ω⋅= js , donde ω es la frecuencia 
en radianes por segundo. 
 
Anteriormente teníamos, que: 
 

 
L

tiRtV
dt

tdi SS )()()( ⋅−=       ( 3 ) 

 
Que se puede escribir de la forma: 
 

 
L

siRsV
sis SS )()(
)(

⋅−
=⋅       ( 5 ) 

 
Obteniendo: 
 

 
SRLs

sVssi
+⋅

= )(
)(        ( 6 ) 

 
 
Que podemos utilizan en SciLab con el siguiente algoritmo: 
 

// Simulación en el Dominio de la Frecuencia de un circuito RL Serie 
(rl_s.sci). 
 
clear 
 
Vs = 10;      // en Voltios cte en este caso 
 
L = 0.016;    // en Henrios 
Rs = 0.1;     // en Ohmios 
 
f=logspace(0,6); 
w=2*%pi*f; 
s=%i*w; 
i_s=Vs./(s*L+Rs); 
 
Mag_i=abs(i_s);   // Magnitud de i(s) 
scf(0) 
clf() 
subplot(2,1,1) 
plot2d('ln',f,Mag_i) 
xtitle('i(s)','f [Hz]','i [A]') 
 
Ph_i=atan(imag(i_s),real(i_s)); // fase en radianes 
Ph_i=Ph_i/%pi*180;              // en grados 
 
subplot(2,1,2) 
plot2d('ln',f,Ph_i) 
xtitle('','f [Hz]','Phase [º]') 

 Donde podemos apreciar que el circuito empieza a comportarse como un inductor ideal 

a frecuencias superiores a 
L

RS

π2
100 ⋅ , tal y como vimos en el apartado sobre la frecuencia 

mínima de trabajo. A partir de esta frecuencia el desfase entre la corriente del inductor y su 
tensión, se aproxima a -90º, dato que nos indica el inductor se comporta como tal, se decir, el 
efecto inductivo predomina sobre el resistivo, tal y como queríamos, para minimizar las 
pérdidas de potencia y calentamiento debido a la resistencia del hilo de cobre utilizado en el 
bobinado. 
 
También podemos determinar )()( sisLsVL ⋅⋅= , expresión obtenida de la conversión al 

dominio de la frecuencia de 
dt

tdiLtVL
)(

)( = . 

Obteniendo este otro algoritmo: 
 

s=%i*w; 
VL_s=L*s.*i_s; 
 
Mag_VL=abs(VL_s);   // Magnitud de i(s) 
scf(1) 
clf() 
subplot(2,1,1) 
plot2d('ln',f,Mag_VL) 
xtitle('VL(s)','f [Hz]','VL [V]') 
 
Ph_VL=atan(imag(VL_s),real(VL_s)); // fase en radianes 
Ph_VL=Ph_VL/%pi*180;              // en grados 
 
subplot(2,1,2) 
plot2d('ln',f,Ph_VL) 
xtitle('','f [Hz]','Phase [º]') 

Efectos de Rs en el Dominio de la Frecuencia. 
 
Para apreciar los efectos de la variación de la variación de SR  y de L , podemos hacer una 
serie de simulaciones para distintos valores de dichos parámetros como hemos hecho en el 
caso anterior en el dominio del tiempo. 

Efectos de L en el Dominio de la Frecuencia. 

Modelo con Características Geométricas. 
 
En este capítulo se describe el desarrollo matemático que permite analizar y simular el 
comportamiento de nuestro inductor, con un modelo un poco más complicado, ya que ahora 
tendremos en cuenta el número de espiras, la permeabilidad magnética, y las dimensiones del 
material ferromagnético con el que se ha construido el inductor.  
 

  
 
 
Ahora tenemos que la inductancia L , de nuestro inductor viene determinada por, tres variables 
nuevas, N es el número de espiras, ML  es la longitud media del circuito magnético y MA  es la 
sección del circuito magnético. 
 
Siendo MA  i ML  variables que podemos determinar fácilmente a partir de las características 
geométricas del material ferromagnético utilizado en el diseño de nuestro inductor: 
 
 ( )ieM rrAlturaA −⋅=  
 

 




 +

=
2

2 ie
M

rr
L π  

 
Conociendo las nuevas variables, veamos como introducimos estos parámetros en nuestro 
modelo matemático, para poder simular nuestro inductor. 
 
En un inductor tenemos una tensión inducida proporcional a variación del flujo magnético en su 
interior y al número de espiras: 
 

 
dt

tdNVind
)(Φ=  

Y como por teoría electromagnética tenemos que: 
 
 MLtiNtH /)()( ⋅=  
 )()( tHtB µ=  

 )()( tBAt M=Φ  
 

ri 
re 

re-ri 

Altura 



 15 

Tenemos que: 

dt
tdi

L
NA

dt
tdHNA

dt
tdBNA

dt
tdNtV

M

M
MML

)()()()(
)(

2µµ ===Φ=   ( 7 ) 

 
En el anterior modelo teníamos: 

dt
tdiLtVL
)(

)( =  

 
Por lo que podemos deducir la descripción matemática de L en función de MM LA ,,µ y N : 

 
M

M

L
NAL

2µ=          ( 8 ) 

 
Por tanto, sólo tenemos que utilizar esta expresión para determinar el valor de la inductancia, 
para poder utilizar el mismo algoritmo de simulación anterior para la simulación del inductor con 
los nuevos parámetros. 
 
El nuevo algoritmo para SciLab es el siguiente (rl_g.sci) :  
 

// Simulación Inductor con Características Geométricas ( rl_g.sci). 
 
clear 
 
N=100;   // Número de espiras. 
 
Am=97.5e-6;   // Sección circuito magnético en m2. 
Lm=105e-3;    // Longitud media del circuito magnético en m. 
u=7e-4;       // Permeabilidad del circuito magnético en H/m. 
 
L = Am*u*N*N/Lm;    // en Henrios 
 
// Variación de L en función de N 
 
VarN=linspace(0,2*N); 
scf(0) 
clf() 
VarL=Am*u*VarN.*VarN/Lm; 
plot(VarN,VarL) 
xtitle('L(N)','N','L [H]') 
 
Rs = 0.1;       // en Ohms 
 
Vs = 10;       // en Voltios 
 
Tau = L/Rs; 
tfin = 6*Tau;     // en s 
dt = Tau/10;    // en s 
 
i(1) = 0;     // Condiciones iniciales 
t(1) = 0;   // Condiciones iniciales 
VL(1)= Vs;    // per t=0, i=0, VR=0 => VL=Vs 
 
for n = 2:tfin/dt 
  t(n) = t(n-1)+dt; 
  i(n) = (Vs*dt+L*i(n-1))/(L+Rs*dt); 
  Vr(n)= i(n)*Rs; 
  VL(n)= L * (i(n)-i(n-1))/dt; 
end; 
 
scf(1) 
clf() 
subplot(2,1,1) 

plot(t,i) 
xtitle('','t [s]','Current [A]') 
 
subplot(2,1,2) 
plot(t,VL) 
xtitle('','t [s]','VL [V]') 
 
f=logspace(-1,4); 
w=2*%pi*f; 
i_s=Vs./(%i*w*L+Rs); 
 
Mag_i=abs(i_s);   // Magnitud de i(s) 
scf(2) 
clf() 
subplot(2,1,1) 
plot2d('ln',f,Mag_i) 
xtitle('i(s)','f [Hz]','i [A]') 
 
Ph_i=atan(imag(i_s),real(i_s)); // phase en radianes 
Ph_i=Ph_i/%pi*180;              // en grados 
 
subplot(2,1,2) 
plot2d('ln',f,Ph_i) 
xtitle('','f [Hz]','Phase [º]') 

 
 
Este algoritmo nos permite apreciar la variación de la Inductancia L , en función del número de 
espiras N , del bobinado de nuestro inductor: 

El comportamiento de nuestro inductor a la aplicación de una tensión constante de 10 V. 

Y el comportamiento de nuestro inductor a diferentes frecuencias de trabajo. 

Conclusiones. 
En este documento se ha presentado un procedimiento sencillo y simplificado de simulación de 
inductores, que nos proporciona una buena base para obtener simulaciones más refinadas y 
precisas. Simulaciones en las cuales tendremos en cuenta la no linealidad del material 
ferromagnético. Es decir el responsable de que la inductancia L no sea constante. Ya que dicha 
inductancia depende de la permeabilidad del material ferromagnético utilizado, y dicha 
permeabilidad depende de las condiciones de trabajo, tanto de la corriente como de la 
frecuencia. 
 
Por este motivo, es importante simular nuestro inductor en las mismas condiciones de trabajo 
que requiere nuestra aplicación. Ya que al no considerar la saturación del material 
ferromagnético puede que nuestro inductor no se comporte como esperamos. 
 
Estas dos nuevas variables introducidas, nos complicarán un poco, tanto el diseño como el 
modelo matemático utilizado para la simulación del comportamiento de nuestro inductor. 
 
Espero que este texto os sea de utilidad, y os permita diseñar inductores con el mínimo 
esfuerzo y con unas máximas garantías de éxito. 

Anexo de Algoritmos. 

Efectos de Rs, en Corriente Alterna, rl_Rs.sci. 
 
// Simulación Inductor ( Circuito RL Serie ). 
// Variación de Rs ( rl_Rs.sci ). 
 
clear 
 
Vp = 10;      // Tensión de Pico en Voltios 
fr = 50;      // Frecuencia en Hercios 
pi = %pi;     // Cte 
w = 2*pi*fr   // Radianes / Segundo 
 
L = 0.016;    // en Henrios 
Rs = [0.02,0.1,0.5];     // en Ohms 
 
Tr = 1/fr;        // Periodo señal de entrada 
NumT = 10;         // Número de periodos a simular 
tfin = Tr*NumT;   // en s 
NumP = 20;        // Puntos por periodo a simular 
dt = Tr/NumP;       // en segundos 
 
for m = 1:length(Rs) 
   
  i(m,1) = 0;     // Condiciones iniciales 
  t(m,1) = 0;     // Condiciones iniciales 
  VL(m,1)= 0;    // per t=0, i=0, VR=0 => VL=Vs 
  Vr(m,1)= 0; 
  Pi(m,1)=0; 
 
  for n = 2:tfin/dt 
    t(n) = t(n-1)+dt; 
    Vs(n) = Vp*sin(w*t(n)); 
    i(m,n) = (Vs(n)*dt+L*i(m,n-1))/(L+Rs(m)*dt); 
    Vr(m,n)= i(m,n)*Rs(m); 
    VL(m,n)= L*(i(m,n)-i(m,n-1))/dt; 
    Pi(m,n)= i(m,n)*Vs(n); 
  end 
end 
 
scf(0) 
clf() 
subplot(3,1,1) 
plot(t,VL) 
xtitle('','t [s]','VL [V]') 
legend('Rs = ' + string(Rs)) 
 
subplot(3,1,2) 
plot(t,i) 
xtitle('','t [s]','Current [A]') 
legend('Rs = ' + string(Rs)) 
 
subplot(3,1,3) 
plot(t,Pi) 
xtitle('','t [s]','Pi [W]') 
legend('Rs = ' + string(Rs)) 

Efectos de L, en Corriente Alterna, rl_L.sci. 
 
// Simulación Inductor ( Circuito RL Serie ). 
// Variación de L ( rl_L.sci ). 
 
clear 
 
Vp = 10;      // Tensión de Pico en Voltios 
fr = 50;      // Frecuencia en Hercios 
pi = %pi;     // Cte 
w = 2*pi*fr   // Radianes / Segundo 
 
L = [0.008,0.016,0.032];    // en Henrios 
Rs = 0.1;     // en Ohms 
 
Tr = 1/fr;        // Periodo señal de entrada 
NumT = 10;         // Número de periodos a simular 
tfin = Tr*NumT;   // en s 
NumP = 20;        // Puntos por periodo a simular 
dt = Tr/NumP;       // en segundos 
 
for m = 1:length(L) 
   
  i(m,1) = 0;     // Condiciones iniciales 
  t(m,1) = 0;     // Condiciones iniciales 
  VL(m,1)= 0;    // per t=0, i=0, VR=0 => VL=Vs 
  Vr(m,1)= 0; 
  Pi(m,1)=0; 
 
  for n = 2:tfin/dt 
    t(n) = t(n-1)+dt; 
    Vs(n) = Vp*sin(w*t(n)); 
    i(m,n) = (Vs(n)*dt+L(m)*i(m,n-1))/(L(m)+Rs*dt); 
    Vr(m,n)= i(m,n)*Rs; 
    VL(m,n)= L(m)*(i(m,n)-i(m,n-1))/dt; 
    Pi(m,n)= i(m,n)*Vs(n); 
  end 
end 
 
scf(0) 
clf() 
subplot(3,1,1) 
plot(t,VL) 
xtitle('','t [s]','VL [V]') 
legend('L = ' + string(L)) 
 
subplot(3,1,2) 
plot(t,i) 
xtitle('','t [s]','Current [A]') 
legend('L = ' + string(L)) 
 
subplot(3,1,3) 
plot(t,Pi) 
xtitle('','t [s]','Pi [W]') 
legend('L = ' + string(L)) 

Efectos de Rs, en el Dominio de la Frecuencia, rl_s_Rs.sci. 
 
// Simulación en el Dominio de la Frecuencia de un circuito RL Serie. 
// Variación de Rs (rl_s_Rs.sci). 
 
clear 
 
Vs = 10;      // en Voltios cte en este caso 
 
L = 0.016;    // en Henrios 
Rs = [0.02,0.1,0.5];     // en Ohmios 
 
f=logspace(0,6); 
w=2*%pi*f; 
s=%i*w; 
for m = 1:length(Rs) 
  i_s(m,:)=Vs./(s*L+Rs(m)); 
end 
 
Mag_i=abs(i_s);   // Magnitud de i(s) 
scf(0) 
clf() 
subplot(2,1,1) 
plot2d('ln',f,Mag_i') 
xgrid() 
xtitle('i(s)','f [Hz]','i [A]') 
legend('Rs = ' + string(Rs)) 
 
Ph_i=atan(imag(i_s),real(i_s)); // fase en radianes 
Ph_i=Ph_i/%pi*180;              // en grados 
 
subplot(2,1,2) 
plot2d('ln',f,Ph_i') 
xgrid() 
xtitle('','f [Hz]','Phase [º]') 
legend('Rs = ' + string(Rs)) 
 
s=%i*w; 
for m = 1:length(Rs) 
  VL_s(m,:)=L*s.*i_s(m,:); 
end 
 
Mag_VL=abs(VL_s);   // Magnitud de i(s) 
scf(1) 
clf() 
subplot(2,1,1) 
plot2d('ln',f,Mag_VL') 
xgrid() 
xtitle('VL(s)','f [Hz]','VL [V]') 
legend('Rs = ' + string(Rs)) 
 
Ph_VL=atan(imag(VL_s),real(VL_s)); // fase en radianes 
Ph_VL=Ph_VL/%pi*180;              // en grados 
 
subplot(2,1,2) 
plot2d('ln',f,Ph_VL') 
xgrid() 
xtitle('','f [Hz]','Phase [º]') 
legend('Rs = ' + string(Rs)) 

Efectos de L en el Dominio de la Frecuencia, rl_s_L.sci. 
 
// Simulación en el Dominio de la Frecuencia de un circuito RL Serie. 
// Variación de L (rl_s_L.sci). 
 
clear 
 
Vs = 10;      // en Voltios cte en este caso 
 
L = [0.008,0.016,0.032];    // en Henrios 
Rs = 0.1;     // en Ohmios 
 
f=logspace(0,6); 
w=2*%pi*f; 
s=%i*w; 
for m = 1:length(L) 
  i_s(m,:)=Vs./(s*L(m)+Rs); 
end 
 
Mag_i=abs(i_s);   // Magnitud de i(s) 
scf(0) 
clf() 
subplot(2,1,1) 
plot2d('ln',f,Mag_i') 
xgrid() 
xtitle('i(s)','f [Hz]','i [A]') 
legend('L = ' + string(L)) 
 
Ph_i=atan(imag(i_s),real(i_s)); // fase en radianes 
Ph_i=Ph_i/%pi*180;              // en grados 
 
subplot(2,1,2) 
plot2d('ln',f,Ph_i') 
xgrid() 
xtitle('','f [Hz]','Phase [º]') 
legend('L = ' + string(L)) 
 
s=%i*w; 
for m = 1:length(L) 
  VL_s(m,:)=L(m)*s.*i_s(m,:); 
end 
 
Mag_VL=abs(VL_s);   // Magnitud de i(s) 
scf(1) 
clf() 
subplot(2,1,1) 
plot2d('ln',f,Mag_VL') 
xgrid() 
xtitle('VL(s)','f [Hz]','VL [V]') 
legend('L = ' + string(L)) 
 
Ph_VL=atan(imag(VL_s),real(VL_s)); // fase en radianes 
Ph_VL=Ph_VL/%pi*180;              // en grados 
 
subplot(2,1,2) 
plot2d('ln',f,Ph_VL') 
xgrid() 
xtitle('','f [Hz]','Phase [º]') 
legend('L = ' + string(L)) 

